

```

1      {
      TITLE:   Олимпиадная информатика

      File:    bliz.pas

      Задача.  Антон в школе начал изучать математику. Его внимание
                привлекло новое для него понятие числовой прямой. Антон
                быстро научился вычислять расстояния между двумя точками
                на этой числовой прямой, задавать отрезки и интервалы на
                ней. Готовясь к контрольной работе, Антон столкнулся со
                следующей задачей: "На числовой прямой задано N точек.
                Необходимо найти среди них две ближайшие". Расстояние
                между двумя точками числовой прямой x и y равно  $|x - y|$ .
                Первая строка входного файла INPUT.TXT содержит
                количество точек N ( $2 \leq N \leq 10^5$ ).

                Вторая строка содержит N различных целых чисел –
                координаты точек на прямой. Числа в строке разделены
                пробелом. Значения всех координат не превосходят  $10^9$  по
                абсолютной величине.

                В первой строке выходного файла OUTPUT.TXT необходимо
                вывести минимальное расстояние между двумя точками,
                заданными во входном файле. Во второй строке выходного
                файла необходимо вывести номера точек, которым
                соответствует найденное расстояние. Точки нумеруются
                натуральными числами от 1 до N в том порядке, в котором
                они заданы во входном файле. Если ответов несколько,
                вывести тот из них, в котором точки расположены левее
                других на числовой прямой. Первым выводится номер левой
                точки, далее через пробел – номер правой точки.

                -----
                INPUT.TXT                                OUTPUT.TXT
                5                                         1
                10 3 6 2 5                                4 2
                -----

      Решение.  Калмыков Вадим (ProCrypt),
                г. Сургут, ЦНИТ "Северная Звезда",
                15:21, 03.06.2008

      Source :  http://acm.dvpion.ru
                (Окружная олимпиада по информатике, 2007–2008)

      Edit.    MSP, 02.12.09, 11:48, (c) LIST2008
    }

```

```

41  Type
42      Zapis = Record
43          X : LongInt; { координата точки на прямой }
44          N : LongInt; { ее порядковый номер в последовательности }
45      end;

46  Const
47      Size = 100000;

48  Var
49      i,          { счетчик циклов }
50      N : LongInt; { число точек }

51      min : LongInt; { минимальное расстояние между точками }

52      x, y : LongInt; { координаты искоемых точек }

53      A : array [1..Size] of Zapis; { массив точек }

54      { Процедура сортировки массива методом Пирамидальной сортировки }
55  Procedure PyramSort;
56  var
57      i, t : LongInt; { счетчики циклов }

58      w : Zapis; { переменная для обмена значений массива }

59  Procedure Pyram (r, q : LongInt);
60  var
61      {
62          r - номер проталкиваемого элемента, q - граница массива
63      }

64      v : Zapis;
65      i, j : LongInt;
66      flag : Boolean;

67  begin
68      i := r;
69      v := A [i]; { индекс рассматриваемого элемента и сам элемент }

70      j := 2 * i; { индекс элемента, с которым происходит сравнение }

71      flag := False; { считаем, что для элемента не найдено место в
72                      пирамиде }

73      While (j <= q)
74          and
75          not (flag)
76      do
77          begin

```

```

78     if (j < q)
79         then
80             if (A [j].X > A [j + 1].X)
81                 then
82                     Inc (j); { сравниваем с меньшим элементом }

83     if (v.X <= A [j].X)
84         then
85             flag := True { элемент находится на своем месте }
86         else
87             begin
88                 A [i] := A [j];
89                 i := j;
90                 j := 2 * i; { переставляем элемент и идем дальше по
                               пирамиде }

92             end;

93     A [i] := v;
94 end;
95 end;

96 begin
97     { последние (N - t) элементов образуют пирамиду }
98     t := N div 2 + 1;

99     for i := (t - 1) downto 1 do
100         Pyram (i, N); { строим пирамиду с конца массива }

101     for i := N downto 2 do
102         begin
103             w := A [1];      { меняем местами 1-й и i-й элементы }
104             A [1] := A [i];
105             A [i] := w;

106             Pyram (1, i - 1); { проталкиваем 1-й элемент }
107         end;
108     end;

109 Begin
110     Assign (Input, 'input.txt');
111     Assign (Output, 'output.txt');

112     Reset (Input);
113     ReadLn (N);

114     { заполняем массив данными из входного файла }
115     for i := 1 to N do
116         begin
117             Read (A [i].X);
118             A [i].N := i;
119         end;

```

```

120     Close (Input);

121     PyramSort; { Сортируем массив в порядке невозрастания }

122     Rewrite (Output);

123     { ищем нужные точки: вычисляем расстояние между ближайшими }
124     x := A [1].N;
125     y := A [N].N;
126     { можно обойтись без модуля - знаем знак }
127     min := A [1].X - A [N].X;

128     for i := N downto 2 do
129         if (A [i - 1].X - A [i].X < min)
130             then
131                 begin
132                     min := A [i - 1].X - A [i].X;
133                     x := A [i].N;
134                     y := A [i - 1].N;
135                 end;

136     WriteLn (min);
137     Write (x, ' ', y);

138     Close (Output);
139     End.

```

Listing данной задачи опубликован в сети Internet по адресу
<http://www.Best-Listing.ru/color-10-task-624.html>

Sergey Mitrofanov, 03.09.14, 14:35

E-mail: infostar@mail.ru

© <http://www.Best-Listing.ru/>, 2006–2014